Here is a interactive coordinate geometry program for the HP-41CX I wrote sometime in the 1980s.

```
LBL "C6X"
RCLPTA
LBL e
CLX
SEEKPT
X<>F
SF 27
SF 21
DEG
"EA LN RS LS"
PROMPT
LBL A
"PTANE?"
PROMPT
X<>Z
STO 07
SF 04
GTO 80
LBL b
FC?C 00
SF 00
LBL B
FS? 05
GTO 10
"ST PT?"
PROMPT
SF 05
CF 00
GTO 40
LBL 10
CF 22
"#"
FIX 0
ARCL 00
"L-?"
FIX 4
PROMPT
FC?C 22
GTO e
X<0?
SF 03
ABS
STO 07
"B1?"
PROMPT
FC?C 22
GTO 11
XEQ 09
STO 03
"D1?"

PROMPT
FC?C 22
GTO 12
GTO 50
LBL 11
"D1?"
PROMPT
FC?C 22
GTO 20
STO 05
"D2(-CC)?"
PROMPT
STO 06
X<0?
SF 06
XEQ 06
X↑2
RCL 05
X↑2
+
RCL 06
X↑2
×
2
/
RCL 05
/
RCL 10
/
ACOS
FC?C 06
CHS
+
RCL 05
GTO 50
LBL 12
"B2?"
PROMPT
FC?C 22
GTO 13
XEQ 09
STO 04
XEQ 06
X<>Y
RCL 04
—
SIN
*
RCL 03
STO Z

RCL 04
—
SIN
/
GTO 50
LBL 13
"D2(-SH)?"
PROMPT
STO 06
X<0?
SF 06
XEQ 06
RCL 03
STO T
ST- Z
RDN
P-R
X<>Y
X↑2
RCL 06
X↑2
X<>Y
—
SQRT
FS?C 06
CHS
+
LBL 50
P-R
X<>Y
RCL 02
+
X<>Y
RCL 01
+
RCL 07
LBL 80
XEQ 08
RDN
SAVEX
RDN
SAVEX
RCL 07
FS?C 04
GTO 40
LBL 20
XEQ 08
GET X
GET X
RCL 02

—
X<>Y
RCL 01
—
R-P
STO 10
FS?C 01
RTN
X<>Y
360
MOD
STO 09
"#"
FIX 0
ARCL 00
"L--"
ARCL 07
AVIEW
COS
"N"
X<0?
"S"
RCL 09
SIN
X>0?
"LE"
X<=0?
"LW"
ASIN
ABS
FS? 09
"AZ="
FS? 09
RCL 09
HMS
FIX 4
ARCL X
AVIEW
"D="
ARCL 10
AVIEW
ADV
RCL 07
LBL 40
FC? 00
STO 00
"#"
FIX 0
ARCL X
AVIEW

FIX 4
XEQ 08
GET X
FC? 00
STO 01
GET X
FC?C 00
STO 02
"N="
ARCL Y
AVIEW
"E="
ARCL X
AVIEW
ADV
FS? 07
RTN
FC? 05
GTO e
FS? 00
GTO 00
FS?C 03
GTO 90
FC?C 02
GTO B
RCL 08
STO 07
GTO 20
LBL 00
CF 02
CF 03
GTO B
LBL 09
X<>Y
HR
GTO IND Y
LBL 04
180
—
LBL 02
CHS
LBL 03
180
+
LBL 01
CF 09
RTN
LBL 05
SF 09
RTN

LBL 08
2
*
SEEKPT
RTN
LBL 06
"DP?"
PROMPT
STO 08
SF 01
SF 02
GTO 20
LBL 90
"POC?"
PROMPT
STO 07
CF 22
"Δ?"
PROMPT
HR
FS?C 22
GTO 14
"L?"
PROMPT
RCL 10
/
R-D
LBL 14
STO Y
RCL 09
180
+
+
STO 04
RDN
HMS
"Δ="
ARCL X
AVIEW
LAST X
D-R
RCL 10
*
"L="
ARCL X
AVIEW
ADV
RCL 04
RCL 10
GTO 50

LBL D
SF 07
"OC PT?"
PROMPT
XEQ 40
SF 00
LBL E
"P1↑P2?"
PROMPT
E3
/
+
STO 07
SF 07
LBL 07
RCL 07
FS? 00
XEQ 20
FC? 00
XEQ 40
ISG 07
GTO 07
GTO e
END
```

324 steps
596 bytes

To run, type in data file name in Alpha register, then press XEQ ALPHA CGX ALPHA

* Number of pts you can store is $\frac{1}{2}$ × length of file (e.g. a file sized with 50 regs holds 25 pts.) Pt. numbering starts at 0 (zero).

Menu: EA LN RS LS (A B D E)

EA ⇒ enter & assign
LN ⇒ lines
RS ⇒ radial stakeout
LS ⇒ list

Entering Bearings: Type in Bearing Quad ENTER↑ R/S

Quad codes are
1 = NE
2 = SE
3 = SW
4 = NW
5 = NA

(Once an azimuth has been entered, the computer will always display azimuths until a bearing is entered, and vice-versa)

Sideshots: Press [f] [D] to toggle between "sideshot" & "normal" mode. You are in sideshot mode when the little "0" is on the display.

1) when you exit the lines routine, SS mode is turned off
2) when you restart at a new point, SS mode is turned off
3) In sideshot mode, curves are not computed.
4) When you execute a radial stakeout, SS mode is turned on.
5) You may sideshot intersections.

Curves: To traverse curves, enter the radius point as a negative number. (Notice the little "3" comes on). Traverse or inverse to the radius point, enter the P.O.C. then give either the Δ or length (If the curve turns left, Δ and L are negative values.)

Errors: If you make a serious mistake, or see a message like "DATA ERROR", press [f] [C] to reset the computer and return to the menu.

Abbreviations:

ST PT	starting point	NE	northeast	N	northing
B1	bearing 1	NW	northwest	E	easting
B2	bearing 2	SE	southeast	[f] [Δ]	delta
D1	distance 1	SW	southwest	L	length
D2	distance 2	AZ	azimuth	∝ PT	occupied point
CC	counterclockwise	D	distance	DP	destination point
SH	short	PT	point		

TOAL ENGINEERING
139 Navarro
SAN CLEMENTE, CA 92672
Phone 492-8586

JOB _____

SHEET NO. 1 OF _____

CALCULATED BY _____ DATE _____

CHECKED BY _____ DATE _____

SCALE _____

Registers

00	Current Pt #
01	Current N
02	Current E
03	Brg 1
04	Brg 2
05	Dist 1
06	Dist 2
07	Next Pt #
08	Dest Pt #
09	Ref Az
10	Ref Dist

Flags

00	sideshot
01	Intersection Leg 1
02	Intersection Leg 2
03	Curve
04	List or Stakeout loop
05	Lines Routine
06	Short R-D or CC D-D
07	-
08	-
09	Display Azimuth
21	Printer on
22	Numeric entry
27	User mode

Keyboard

A	Enter & Assign
B	Lines
D	Radial Stakeout
E	List
b	Sideshot / Normal mode toggle
e	Error Reset

Local Labels

10	Lines Routine
11	B1 unknown
12	B1 known, D1 unknown
13	B1 known, D1 unknown, B2 unknown
50	Traverse
80	Store
20	Inverse
40	Display
09	Compute Azimuth
04	Q = NW
02	Q = SE
03	Q = SW
01	Q = NE
05	Q = NA
08	Set Data Pointer
06	Intersection Setup
90	Curve
14	Radial Brg out
07	List / Stakeout loop

TOAL ENGINEERING
139 Navarro
SAN CLEMENTE, CA 92672
Phone 492-8586

JOB _____

SHEET NO. 2 OF _____

CALCULATED BY _____ DATE _____

CHECKED BY _____ DATE _____

SCALE _____

Formulas



Brg → Az : Q1 : Az = Brg
Q2 : Az = 180 − Brg
Q3 : Az = 180 + Brg
Q4 : Az = 360 − Brg

Az → Brg : Brg = $|\sin^{-1}(\sin Az)|$

Quad code :

North	Cos(Az) = +	NE	cos +, sin +
South	Cos(Az) = −	SE	cos −, sin +
West	Sin(Az) = −	SW	cos −, sin −
East	Sin(Az) = +	NW	cos +, sin −

Traverse, Inverse



φ = Azimuth from 1 to 2

$\Delta N = d \cos \phi$ $\quad d = \sqrt{\Delta N^2 + \Delta E^2}$

$\Delta E = d \sin \phi$ $\quad \tan \phi = \Delta E / \Delta N$

Use HP-41 [P→R] and [R→P] functions

θ	φ	P→R	ΔE	y
r	d	R→P	ΔN	x

Traverse: N2 = N1 + ΔN
E2 = E1 + ΔE

Inverse: ΔN = N2 − N1
ΔE = E2 − E1

TOAL ENGINEERING
139 Navarro
SAN CLEMENTE, CA 92672
Phone 492-8586

JOB _____
SHEET NO. 3 OF _____
CALCULATED BY _____ DATE _____
CHECKED BY _____ DATE _____
SCALE _____

Bearing - Bearing:



need pt #2

ϕ = ref Az c = ref dist
α = Az1 − ϕ
β = ϕ − Az2
γ = 180 − α + β

Law of sines: $\frac{a}{\sin\beta} = \frac{c}{\sin\gamma}$

$a = \frac{c \sin\beta}{\sin\gamma}$

since $\sin(180-\theta) = \sin\theta$

$a = \frac{c\sin\beta}{\sin(\alpha+\beta)} = \frac{c\sin(\phi - Az2)}{\sin(Az1 - \phi + \phi - Az2)}$

$= \frac{c\sin(\phi - Az2)}{\sin(Az1 + Az2)}$

Note: You can specify Brg1 & Brg2 "forward" or "backward" since $\sin(180-\theta) = \sin\theta$

Bearing - Distance:



θ = ref Az c = ref dist
d = known second distance
$\phi = \theta - Az1$

$a = c\cos\phi$
$h = c\sin\phi$
$b = \sqrt{d^2 - h^2}$

for long solution use a+b (CF 06)
" short " " a−b (SF 06)

Note: ϕ can be specified as either θ − Az1 or Az1 − θ since the sign of ϕ is unimportant
since $a = c\cos\phi = c\cos(-\phi)$
and $h^2 = (c\sin\phi)^2 = (-c\sin(-\phi))^2$
Also, the sign of d is unimportant since it is only used as $d^2 = -d^2$

TOAL ENGINEERING
139 Navarro
SAN CLEMENTE, CA 92672
Phone 492-8586

JOB _____
SHEET NO. 4 OF _____
CALCULATED BY _____ DATE _____
CHECKED BY _____ DATE _____
SCALE _____

Distance - Distance:



θ = ref azimuth c = ref distance
a = known first distance
b = known second distance

Law of cosines:

$$b^2 = a^2 + c^2 - 2ac\cos\phi$$

$$\phi = \cos^{-1}\left(\frac{a^2 + c^2 - b^2}{2ac}\right)$$

for clockwise soln (CF 06) use $Az1 = \theta - \phi$
" counterclockwise " (SF 06) " $Az1 = \phi - \theta$

Curves:



$$Az2 = \Delta + (Az1 + 180)$$

$$L = \Delta_{rad} \cdot r$$

$$\Delta_{rad} = \frac{L}{r}$$

For curves turning to the left Δ or L is specified as a negative value

This is a sketch I used during development of the program